

Appendix 1. Directed Acyclic graphs

Below we show the directed acyclic graphs (DAGs) of the two models specified above. To prevent the graphs from becoming too cluttered we have omitted a few of the intermediate deterministic nodes. In the graphs i is the (loop) index of the individuals, j of the time points and k of the drugs.

Compartment model

The compartment model for the j th response of the k th drug formulation of the i th individual is given by

$$y_{ijk} = \exp(lKa_{ik} + lKe_{ik} - lCl_{ik}) \frac{[\exp(-e^{lKe_{ik}} t_{ijk}) - \exp(-e^{lKa_{ik}} t_{ijk})]}{e^{lKa_{ik}} - e^{lKe_{ik}}} + \epsilon_{ijk}, \quad (1)$$

where y_{ijk} be the k th response at the time t_{ijk} of individual i (for each $i = 1, \dots, n$, $j = 1, \dots, m_i$ and $k = 1, \dots, K$); $lKa_{ik} = lKa_k + \gamma_{i1k}$, $lKe_{ik} = lKe_k + \gamma_{i2k}$ and $lCl_{ik} = lCl_k + \gamma_{i3k}$; lKa_k , lKe_k and lCl_k are the fixed-effects related to the k -th response; γ_{i1k} , γ_{i2k} and γ_{i3k} are the respective random effects. Here, $\gamma_i = (\gamma_{i1}^\top, \dots, \gamma_{iK}^\top)^\top$ and $\gamma_{ik} = (\gamma_{i1k}, \gamma_{i2k}, \gamma_{i3k})^\top$ is the vector of random effects related to the k -th response of the i -th individual for each $k = 1, \dots, K$ and $i = 1, \dots, n$. We model $\gamma_i = b_i^T \xi$ where $b_i \sim N(0, \Sigma_b)$ and the elements of x_i are i.i.d. and uniformly distributed on $[0, 100]$. The residual random errors $\epsilon_{ijk} \sim N(0, \sigma_k^2)$ are assumed to be independent of each other and of the random effects.




Figure 1: Directed acyclic graph for the compartment model. Hyperparameters were omitted.

MSITAR model

The multivariate MSITAR model is given by:

$$\begin{aligned} y_{ijk} | \gamma_{Mi}, \beta_k, \sigma_k^2 &\sim N(\exp(\gamma_{i2k})[T_{ijk}^\top \beta_k], \sigma_k^2), \\ T_{ijk} &= B(\exp(\gamma_{i3k})[t_{ijk} + \min(\gamma_{i1k}, 0)]), \end{aligned} \quad (2)$$

where y_{ijk} be the k th response at the time t_{ijk} of individual i . T_{ijk} is a matrix with bases of cubic splines for the k th response of the individual i at time j , $\beta_k = (\beta_{k1}, \dots, \beta_{kL})^T$ is the vector of spline coefficients related to the k -th response (here L denotes the degrees of freedom of the spline), $\gamma_{ik} = (\gamma_{i1k}, \gamma_{i2k}, \gamma_{i3k})^\top$ is the vector of random-effects for $i = 1, \dots, n$; $j = 1, \dots, m$. We model $\gamma = b_i \xi$ with $b_i \sim N(0, \Sigma_b)$ and the elements of ξ are uniformly distributed on $[0, 100]$.

$$\begin{aligned} y_{ijk} | \gamma_{Mi}, \beta_k, \sigma_k^2 &\sim N(\exp(\gamma_{i2k})[T_{ijk}^\top \beta_k], \sigma_k^2), \\ T_{ijk} &= B(\exp(\gamma_{i3k})[t_{ijk} + \min(\gamma_{i1k}, 0)]), \end{aligned} \quad (3)$$

where y_{ijk} be the k th response at the time t_{ijk} of individual i . T_{ijk} is a matrix with bases of cubic splines for the k th response of the individual i at time j , $\beta_k = (\beta_{k1}, \dots, \beta_{kL})^T$ is the vector of spline coefficients related to the k -th response (here L denotes the degrees of freedom of the spline), $\gamma_{ik} = (\gamma_{i1k}, \gamma_{i2k}, \gamma_{i3k})^\top$ is the vector of random-effects for $i = 1, \dots, n$; $j = 1, \dots, m$. We model $\gamma = b_i \xi$ with $b_i \sim N(0, \Sigma_b)$ and the elements of ξ are uniformly distributed between 0 and 100.




Figure 2: Directed acyclic graph for the MSITAR model. Hyperparameters were omitted.

Appendix 2. JAGS code

In this appendix we give the JAGS code of the two models that we compared.

2.1 Compartment model

The listing below details the compartment model in JAGS. We called JAGS from R using the jagsUI library (which allows us to run several chains in parallel). As explained in section 3.1 of the main text the gammas and phis are not identified. We therefore define k_e as the lowest of the two random effects and k_a as the highest. Note that a few of the deterministic nodes we use in the program do not occur in the main text. They mainly function to make the program clearer and to prevent some lines from becoming too long. Like in the main article we use $i \in 1 \dots n$ as an index of the individuals, $j \in 1 \dots m$ as an index for the measurement time and $k \in 1 \dots K$ as the index of the drug. $lKa_1, lKe_1, \dots, lCl_2$ were renamed to **beta1[1], beta1[2], ..., beta2[3]** because JAGS does not allow the distinction between different symbols by using different numbers of indices. In this JAGS code m , does not depend on the individual as the data is balanced. Note that the number of drugs K is hard-coded as two. Also the limit of detection is hard-coded.

Listing 1: Compartment model

```
model{
  for(i in 1:n) {
```

```

b[i, 1:6] ~ dnmnorm(mu.b[1:6], tau.b[1:6, 1:6])
for (r in 1:6){
  gammas[i, r] ← b[i, r] * xi[r]
}
for(k in 1:2){
  lKe[i, k] ← min(phi1[i, k], phi2[i, k])
  lKa[i, k] ← max(phi1[i, k], phi2[i, k])
  ke[i, k] ← exp(lKe[i, k])
  ka[i, k] ← exp(lKa[i, k])
}
phi1[i, 1] ← beta1[1] + gammas[i, 1]
phi2[i, 1] ← beta1[2] + gammas[i, 2]
phi3[i, 1] ← beta1[3] + gammas[i, 3]
phi1[i, 2] ← beta2[1] + gammas[i, 4]
phi2[i, 2] ← beta2[2] + gammas[i, 5]
phi3[i, 2] ← beta2[3] + gammas[i, 6]
} # for(i in 1:n)
tau.b[1:6, 1:6] ~ dwish(6 * Omega[ , ], 6)
for(j in 1:m){
  for(i in 1:n){
    for(k in 1:2){
      observed[i, j, k] ~ dinterval(y[i, j, k], 2)
      y[i, j, k] ~ dnorm(mu[i, j, k], tau[k]) #data model: the likelihood
      mu[i, j, k] ← exp(lKe[i, k]+lKa[i, k]-phi3[i, k]) *
        (exp(-ke[i, k]*time[j])-exp(-ka[i, k]*time[j])) /
        (ka[i, k]-ke[i, k])
      res[i, j, k] ← y[i, j, k] - mu[i, j, k]
    } # for(k in 1:2)
  } # for(i in 1:n)
} # for(j in 1:m)
for(k in 1:2){
  tau[k] ~ dgamma(0.001, 0.001)
}
for(u in 1:3){
  beta1[u] ~ dnorm(0, 0.0001)
  beta2[u] ~ dnorm(0, 0.0001)
}
for (r in 1:6){
  xi[r] ~ dunif(0, 100)
}
}

```

2.2 MSITAR model model

Now we present the SITAR model. In this model **Bas** represents the restricted cubic spline basis that is evaluated on a fine grid of values from **R**. This is done by a modified version of the **ns** function from the **R** splines package. From which the lines that apply the restrictions on the spline basis are replaced by the restrictions that are detailed in the paper. This modified **ns** function is available from the GitHub page of the first author (<https://github.com/stenw/FlexibleBioequivalence>). The **interp.lin** func-

tion now interpolates between those values.

Listing 2: SITAR model

```

model{
  #subject specific effects
  tau.b[1:6, 1:6] ~ dwish(6 * Omega[,], 6)
  for (r in 1:6){
    xi[r] ~ dunif(0, 100)
  }
  for(i in 1:n) {
    b[i, 1:6] ~ dmnorm(mu.b[1:6], tau.b[1:6, 1:6])
    for (r in 1:6){
      gammas[i, r] ← b[i, r] * xi[r]
    }
    shift1[i] ← min(gammas[i, 5], 0)
    shift2[i] ← min(gammas[i, 6], 0)
    for(j in 1:m){
      x[i, j, 1] ← exp(gammas[i, 1]) * (time[j] + shift1[i])
      x[i, j, 2] ← exp(gammas[i, 2]) * (time[j] + shift2[i])
      mu[i, j, 1] ← exp(gammas[i, 3]) *
      inprod(beta1[1:L], B[i,j, 1:L,1])
      mu[i, j, 2] ← exp(gammas[i, 4]) *
      inprod(beta2[1:L], B[i,j, 1:L,2])
      for(k in 1:2){
        y[i, j, k] ~ dnorm(mu[i, j, k], tau[k])
        res[i, j, k] ← y[i, j, k] - mu[i, j, k]
        observed[i, j, k] ~ dinterval(y[i, j, k], 2)
        for(l in 1:L){
          B[i, j, l, k] ← interp.lin(x[i, j, k], grid, Bas[, l])
        } # for l
      } # for k
    } # for j
  } # for i
  for(k in 1:2){
    tau[k] ~ dgamma(0.001, 0.001)
  }
  for(l in 1:L){
    beta1[l] ~ dnorm(0, 0.0001)
    beta2[l] ~ dnorm(0, 0.0001)
  }
}

```

Appendix 3. Sampling procedures in JAGS

To obtain samples from the posterior of our models we use the program ‘Just another Gibbs sampler’ (JAGS) by Martyn Plummer (Plummer, 2003). This program can be seen as a reimplementation of WinBUGS in C++, although there are some differences. A big advantage of BUGS-like software, is that it makes sampling from any model relatively easy as one only needs to specify the model and the likelihood, the rest is done by the program. JAGS will then determine the directed acyclic graph, derive the full conditionals and find suitable samplers. The only disadvantage of JAGS or similar software is its black-box nature, i.e. the user is not fully aware how the sampling is done. Below we give details what samplers JAGS has used for the different full conditionals of our models. JAGS converts the program text to a graph of nodes storing information about the distribution of each variable and the relations that exist between them. This graph of nodes is traversed and for each node a suitable sampler is chosen.

The samplers chosen for the MSITAR model are as follows: For the residual variances τ conjugate gamma samplers are used.

$$\tau_k \mid \text{rest} \stackrel{\text{iid}}{\sim} \text{Gamma} \left(\alpha_\tau + \frac{1}{2} \sum_i m_i, \beta_\tau + \frac{1}{2} \sum_{i,j} \{y_{ijk} - \mu_{ijk}\}^2 \right)$$

where n is the number of individuals and m is the number of measurement times so nm is the total number of observations for either drug and

$$\mu_{ijk} = \exp(b_{i2k}\xi_2)T_{ijk}\beta_k$$

is the expected value under the MSITAR model. Here order_i is the covariate that denotes the order in which the drugs are given and α_τ and β_τ are the prior shape and rate parameters for the gamma prior of the τ s. For the fixed effects β conjugate normal samplers are used:

$$\begin{aligned} \beta \mid \text{rest} &\sim N(\bar{\mu}_{kp}, \bar{\tau}_{\beta_{kp}}^{-1}) \\ \text{where } \bar{\mu}_{kp} &= \bar{\tau}_{\beta_{kp}}^{-1} \tau \sum_{ij} A_{ijkp}^2 B_{ijkp}, \\ \bar{\tau}_{\beta_{kp}} &= \tau_{\beta_{kp}} + \sum_{ij} B_{ijkp}^2 \tau_k, \\ A_{ijkp} &= Y_{ijk} - \exp(b_{i2k}) \sum_{l \neq p} \beta_l (T_{ijk})_l, \\ B_{ijkp} &= \exp(b_{i2k}) \beta_l (T_{ijk})_p \end{aligned}$$

with $(T_{ijk})_l$ the l th element of vector T_{ijk} and $\tau_{\beta_{kp}}$ the prior precision of β_{kp} . A_{ijkp} can be thought of as the response when we have removed the effect of

the other β s, while B_{ijkp} is the effect that a unit increase of the coefficient has on this outcome. For the precision of the random effects a conjugate Wishart sampler is used, i.e.

$$T_{\mathbf{b}} \mid \text{rest} \sim W\left(\delta + n, \Omega + \sum_i \mathbf{b}_i \mathbf{b}_i^T\right).$$

where δ is the prior number of degrees of freedom, I the number of individuals and Ω the prior variance matrix.

For the random effects \mathbf{b} , a Metropolis sampler is used. Given the other parameters \mathbf{b}_i is sampled from a distribution proportional to:

$$\exp\left(-\frac{1}{2} \mathbf{b}_i' T_{\mathbf{b}}^{-1} \mathbf{b}_i\right) \prod_{jk} g(y_{ijk} | \beta_k, \tau_k, \mathbf{b}_i, \xi)$$

Here $T_{\mathbf{b}}$ is the precision of the \mathbf{b} s and we use $g(y_{ijk} | \beta_k, \tau_k, \mathbf{b}_i, \xi)$ for the density of the MSITAR model conditional on the random effects, that is:

$$\sqrt{\frac{\tau_k}{2\pi}} \exp\left[-\frac{1}{2}(y_{ijk} - \mu_{ijk})^2 \tau_k\right] \quad (4)$$

For the auxiliary parameters related to the random effect variances ξ and the censored blood concentrations y_{ijk} , slice samplers are used. Given all other parameters ξ is sampled from a distribution proportional to:

$$\prod_{ijk} g(y_{ijk} | \beta_k, \tau_k, \mathbf{b}_i, \xi), \quad (5)$$

Finally given the other parameters we can sample the elements of y_{ijk} that are censored from:

$$\frac{\tau_k \phi(\tau_k(y_{ijk} - \mu_{ijk}))}{\Phi(\tau_k(2 - \mu_{ijk}))}$$

where $\phi(x)$ is the pdf and $\Phi(x)$ is the CDF of a standard normal.

In the parametric model most sampler types are identical to the corresponding ones chosen in the MSITAR model, only for the β 's slice samplers are chosen instead of a conjugate normal sampler as the response is not linear in the fixed effects in this model.

Sampling now begins with an adaptive phase in which samplers that require it, can change their behavior. For the Metropolis sampler this means that JAGS keeps track of a running mean of the acceptance rate, giving a larger weight to more recent iterations, and the variances of the proposal distribution are adjusted to aim for a target acceptance rate of 0.234. For

the slice samplers we similarly keep track of a weighed average of the jumps and the step size (the tuning parameter that controls the initial estimate of the width of the slice) is adjusted accordingly so fewer steps will have to be taken to find the correct slice of the slice. After the adaptive phase we continue the burn-in. When we are convinced we are sampling from the posterior we start storing the sampled values to be used in our further analyses.

Appendix 4. Residual plots

In this section we provide the (marginal) residuals plots of the two models. They are defined as:

$$y_{ijk} - \exp(lKa_{ik} + lKe_{ik} - lCl_{ik}) \frac{[\exp(-e^{lKe_{ik}} t_{ijk}) - \exp(-e^{lKa_{ik}} t_{ijk})]}{e^{lKa_{ik}} - e^{lKe_{ik}}},$$

for the compartment model and as

$$y_{ijk} - (\exp(\gamma_{i2k})[T_{ijk}^\top \boldsymbol{\beta}_k])$$

for the MSITAR model. The heteroskedasticity is clearly visible.




Figure 3: Residual plot for the MSITAR model




Figure 4: Residual plot for the compartment model

References

- Levine, J. and John, L. (2009). *Flex & Bison*. O'Reilly Media, Inc., 1st edition.
- Plummer, M. (2003). Jags: A program for analysis of Bayesian graphical models using Gibbs sampling. In *Proceedings of the 3rd International Workshop on Distributed Statistical Computing*.